1-co middle c program.

The format is E = {Const,Linear,Quad,Pert} which means $O\left(\mathrm{Const}^{-1} e^{\mathrm{Linear}+\frac{\mathrm{Quad}}{\mathrm{Const}}}\left(1+\epsilon \frac{\mathrm{Pert}}{\mathrm{Const}^4}\right)\right)$.

This is a stand alone notebook but, all formulas were checked in NewVdVAlgebraAt17.nb

```
Logos[α_, β_, δ_, μ_ , a_, c_, w_, t_] :=
 4 a c w δ^2 μ^2 + δ (1 + μ) (w^2 α^2 + a^2 β^2 ) +
  a^2 w^2 δ^3 (1 + 3 μ) + (1 - t) (2 (α β + δ μ)^2 - α^2 β^2) +
  2 (α β + 2 δ μ + a w δ^2 (1 + 2 μ) + 2 c δ μ^2) (w α + a β) +
  4 (c μ^2 + a w δ (1 + μ) ) (α β + δ μ)
```

```
(*R-matrix*)
Rp[i_, j_] := {1, Log[t_i] c_j, a_i w_j, a_i c_i w_j + c_i c_j + (a_i^2 w_j^2)/4}
Rm[i_, j_] := {1, -Log[t_i] c_j, -t_i^-1 a_i w_j, -c_i c_j + t_i^-1 a_i c_j w_j - (t_i^-2 a_i^2 w_j^2)/4}
```

```
(*Cuaps*)
nr[i_] := {t_i^(1/2), 0, 0, -c_i t_i^2}
nl[i_] := {1, 0, 0, 0}
ur[i_] := {t_i^(-1/2), 0, 0, c_i t_i^-2}
ul[i_] := {1, 0, 0, 0}
```

```
DP_{x_→D_α_, y_→D_β_}[P_][f_] :=
 Total[CoefficientRules[P, {x, y}] /. ({m_, n_} → c_) :→ c D[f, {α, m}, {β, n}]]
DP_{x_→D_α_, y_→D_β_, z_→D_γ_, w_→D_δ_}[P_][f_] := Total[CoefficientRules[P, {x, y, z, w}] /.
   ({m_, n_, o_, p_} → c_) :→ c D[f, {α, m}, {β, n}, {γ, o}, {δ, p}]]
```

```
(*Stitching operation*)
NOwa[i_, j_, k_][Z_] := Block[
  {α, β, δ, C, L, Q, P, QRest, μ, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = μ^-1 ((1 - t_k) α β + α w_i + β a_j + δ a_j w_i);
  P = DP_{w_i→D_α, a_j→D_β}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α = C^-1 D[Q, w_i] /. a_j → 0;
  β = C^-1 D[Q, a_j] /. w_i → 0;
  δ = C^-1 D[Q, a_j, w_i];
  QRest = Q /. {a_j → 0, w_i → 0};
  μ = 1 - (1 - t_k) δ;

  Q = μ QRest + C * ((1 - t_k) α β + α w_i + β a_j + δ a_j w_i);
  P = μ^4 P - 1/4 (1 + t_k) C^4 Logos[α, β, δ, μ , a_j, c_k, w_i, t_k];
  C = μ C;
  {C, L, Q, P} // Together
 ]

NOc[i_, j_][Z_] := Block[
  {α, β, γ, Γ, δ, C, L, Q, P, QRest, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = e^(Γ+γ) δ a_j w_i + e^Γ α w_i + e^γ β a_j + Γ c_j + γ c_i;
  P = DP_{w_i→D_α, c_j→D_Γ, a_j→D_β, c_i→D_γ}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α = C^-1 D[Q, w_i] /. {a_j → 0};
  β = C^-1 D[Q, a_j] /. {w_i → 0};
  δ = C^-1 D[Q, a_j, w_i];
  γ = D[L, c_i];
  Γ = D[L, c_j];
  QRest = Q /. {a_j → 0, w_i → 0};
  Q = QRest + C e^(Γ+γ) δ a_j w_i + C e^Γ α w_i + C e^γ β a_j;


  {C, L, Q, P} // Expand
 ]

m[i_, j_, k_][Z_] := Block[{x}, (Z // NOwa[i, j, x] // NOc[i, j]) /.
    {a_{i|j|x} → a_k, c_{i|j|x} → c_k, w_{i|j|x} → w_k, t_{i|j|x} → t_k} // Expand]
```

```
ZZ = {10 + t₁, -2 b₂ c₁ + b₁ c₂ + 2 b₃ c₃,
   a₂ w₂ + a₃ w₂ + a₂ w₁ + a₁ w₁ + a₃ w₁ + a₁ w₂, 3 t₂ a₁ w₁ + a₂ c₃ + t₁ c₁ + a₂ c₂ w₂};
ZZ //
  m[
   1,
   2,
   1];

(*Meta-associativity*)
(ZZ // m[1, 2, 1] // m[1, 3, 1] // Expand) -
  (ZZ // m[2, 3, 2] // m[1, 2, 1] // Expand) // Together

{0, 0, 0, 0}

(*Disjoint union of two tangles, multiplies the constant terms,
adds the exponents taking into account the scalings!*)
DisjUnion[L_] := {Times @@ L[[All, 1]], Plus @@ L[[All, 2]],
   (Times @@ L[[All, 1]]) Plus @@ (L[[All, 3]] / L[[All, 1]]),
   (Times @@ L[[All, 1]])^4 Plus @@ (L[[All, 4]] / L[[All, 1]]^4)} // Expand

(*Reidemeister 2*)
DisjUnion@{Rp[1, 2], Rm[3, 4]} // m[1, 3, x] // m[2, 4, y]

{1, 0, 0, 0}

(*Reidemeister 3*)
(DisjUnion@{Rp[1, 2], Rp[3, 4], Rp[5, 6]} // m[1, 3, a] // m[2, 5, b] // m[4, 6, c]) -
  (DisjUnion@{Rp[1, 2], Rp[3, 4], Rp[5, 6]} // m[3, 5, a] // m[1, 6, b] // m[2, 4, c]) //
 Expand
(DisjUnion@{Rm[1, 2], Rm[3, 4], Rm[5, 6]} // m[1, 3, x] // m[2, 5, y] // m[4, 6, z]) -
  (DisjUnion@{Rm[1, 2], Rm[3, 4], Rm[5, 6]} // m[3, 5, x] // m[1, 6, y] // m[2, 4, z]) //
 Expand

{0, 0, 0, 0}

{0, 0, 0, 0}

(*rotated crossings*)
(DisjUnion@{ur[1], Rm[2, 5], nr[3], ur[4], nr[6]} // m[1, 2, 1] // m[1, 3, 1] //
    m[4, 5, 4] // m[4, 6, 4]) -
 (DisjUnion@{ul[1], Rm[2, 5], nl[3], ul[4], nl[6]} // m[1, 2, 1] // m[1, 3, 1] //
    m[4, 5, 4] // m[4, 6, 4])

{0, 0, 0, 0}

(*the oppositely oriented RII *)
DisjUnion[{Rm[1, 2], Rp[3, 4], ur[5], nr[6]}] // m[1, 3, 1] // m[4, 5, 4] // m[4, 2, 4] //
 m[4, 6, 4]

{1, 0, 0, 0}
```

```
(*Kinks*)
```

$\text{Kp}[i\_] := \left\{\sqrt{t_i}, \text{Log}[t_i]\, c_i, \sqrt{t_i}\, a_i\, w_i, -t_i{}^2 c_i + t_i{}^2 c_i^2 + t_i{}^2 a_i\, c_i\, w_i + \frac{1}{4} t_i{}^2 a_i^2 w_i^2\right\}$

$\text{Km}[i\_] := \left\{\frac{1}{\sqrt{t_i}}, -\text{Log}[t_i]\, c_i, -\frac{a_i\, w_i}{t_i{}^{3/2}}, \frac{1}{4\, t_i{}^4}\left(4\, t_i{}^2 c_i - 4\, t_i{}^2 c_i^2 + 4\, t_i\, a_i\, c_i\, w_i - a_i^2 w_i^2\right)\right\}$

```
Clear[i];
DisjUnion[{Rp[1, 4], nr[2], ul[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, i] // Together //
  Expand
DisjUnion[{Rm[4, 1], nr[2], ul[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, i] // Expand //
  Together
DisjUnion[{Rp[4, 1], nl[2], ur[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, i] // Expand
DisjUnion[{Rm[1, 4], nl[2], ur[3]}] // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, i] // Expand //
  Together
```

$\left\{\sqrt{t_i}, \text{Log}[t_i]\, c_i, a_i \sqrt{t_i}\, w_i, -c_i\, t_i^2 + c_i^2\, t_i^2 + a_i\, c_i\, t_i^2\, w_i + \frac{1}{4} a_i^2\, t_i^2\, w_i^2\right\}$

$\left\{\frac{1}{\sqrt{t_i}}, -\text{Log}[t_i]\, c_i, -\frac{a_i\, w_i}{t_i^{3/2}}, \frac{1}{4\, t_i^4}\left(4\, c_i\, t_i^2 - 4\, c_i^2\, t_i^2 + 4\, a_i\, c_i\, t_i\, w_i - a_i^2\, w_i^2\right)\right\}$

$\left\{\sqrt{t_i}, \text{Log}[t_i]\, c_i, a_i \sqrt{t_i}\, w_i, -c_i\, t_i^2 + c_i^2\, t_i^2 + a_i\, c_i\, t_i^2\, w_i + \frac{1}{4} a_i^2\, t_i^2\, w_i^2\right\}$

$\left\{\frac{1}{\sqrt{t_i}}, -\text{Log}[t_i]\, c_i, -\frac{a_i\, w_i}{t_i^{3/2}}, \frac{1}{4\, t_i^4}\left(4\, c_i\, t_i^2 - 4\, c_i^2\, t_i^2 + 4\, a_i\, c_i\, t_i\, w_i - a_i^2\, w_i^2\right)\right\}$

```
(*Inverse Kinks*)
DisjUnion@{Kp[1], Km[2]} // m[1, 2, 1] // Together
```

{1, 0, 0, 0}

```
(*Powers of the positive kink, patterns?!*)
DisjUnion@{Kp[1]} // Together // Expand
DisjUnion@{Kp[1], Kp[2]} // m[1, 2, 1] // Together // Expand
DisjUnion@{Kp[1], Kp[2], Kp[3]} // m[1, 2, 1] // m[1, 3, 1] // Together // Expand
DisjUnion@{Kp[1], Kp[2], Kp[3], Kp[4]} // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] //
  Together // Expand
DisjUnion@{Kp[1], Kp[2], Kp[3], Kp[4], Kp[5]} // m[1, 2, 1] // m[1, 3, 1] // m[1, 4, 1] //
   m[1, 5, 1] // Together // Expand
```

$\left\{\sqrt{t_1},\ \text{Log}[t_1]\, c_1,\ a_1 \sqrt{t_1}\, w_1,\ -c_1 t_1^2 + c_1^2 t_1^2 + a_1 c_1 t_1^2 w_1 + \frac{1}{4} a_1^2 t_1^2 w_1^2\right\}$

$\{t_1,\ 2\,\text{Log}[t_1]\, c_1,\ a_1 t_1 w_1 + a_1 t_1^2 w_1,$
$-2 c_1 t_1^4 + 2 c_1^2 t_1^4 + a_1 c_1 t_1^4 w_1 + 3 a_1 c_1 t_1^5 w_1 + \frac{1}{4} a_1^2 t_1^4 w_1^2 + a_1^2 t_1^5 w_1^2 + \frac{5}{4} a_1^2 t_1^6 w_1^2\}$

$\{t_1^{3/2},\ 3\,\text{Log}[t_1]\, c_1,\ a_1 t_1^{3/2} w_1 + a_1 t_1^{5/2} w_1 + a_1 t_1^{7/2} w_1,\ -3 c_1 t_1^6 + 3 c_1^2 t_1^6 + a_1 c_1 t_1^6 w_1 +$
$3 a_1 c_1 t_1^7 w_1 + 5 a_1 c_1 t_1^8 w_1 + \frac{1}{4} a_1^2 t_1^6 w_1^2 + a_1^2 t_1^7 w_1^2 + \frac{9}{4} a_1^2 t_1^8 w_1^2 + 3 a_1^2 t_1^9 w_1^2 + \frac{9}{4} a_1^2 t_1^{10} w_1^2\}$

$\{t_1^2,\ 4\,\text{Log}[t_1]\, c_1,\ a_1 t_1^2 w_1 + a_1 t_1^3 w_1 + a_1 t_1^4 w_1 + a_1 t_1^5 w_1,$
$-4 c_1 t_1^8 + 4 c_1^2 t_1^8 + a_1 c_1 t_1^8 w_1 + 3 a_1 c_1 t_1^9 w_1 + 5 a_1 c_1 t_1^{10} w_1 + 7 a_1 c_1 t_1^{11} w_1 +$
$\frac{1}{4} a_1^2 t_1^8 w_1^2 + a_1^2 t_1^9 w_1^2 + \frac{9}{4} a_1^2 t_1^{10} w_1^2 + 4 a_1^2 t_1^{11} w_1^2 + \frac{21}{4} a_1^2 t_1^{12} w_1^2 + 5 a_1^2 t_1^{13} w_1^2 + \frac{13}{4} a_1^2 t_1^{14} w_1^2\}$

$\{t_1^{5/2},\ 5\,\text{Log}[t_1]\, c_1,\ a_1 t_1^{5/2} w_1 + a_1 t_1^{7/2} w_1 + a_1 t_1^{9/2} w_1 + a_1 t_1^{11/2} w_1 + a_1 t_1^{13/2} w_1,$
$-5 c_1 t_1^{10} + 5 c_1^2 t_1^{10} + a_1 c_1 t_1^{10} w_1 + 3 a_1 c_1 t_1^{11} w_1 + 5 a_1 c_1 t_1^{12} w_1 + 7 a_1 c_1 t_1^{13} w_1 + 9 a_1 c_1 t_1^{14} w_1 + \frac{1}{4} a_1^2 t_1^{10} w_1^2 +$
$a_1^2 t_1^{11} w_1^2 + \frac{9}{4} a_1^2 t_1^{12} w_1^2 + 4 a_1^2 t_1^{13} w_1^2 + \frac{25}{4} a_1^2 t_1^{14} w_1^2 + 8 a_1^2 t_1^{15} w_1^2 + \frac{33}{4} a_1^2 t_1^{16} w_1^2 + 7 a_1^2 t_1^{17} w_1^2 + \frac{17}{4} a_1^2 t_1^{18} w_1^2\}$

```
(*3_1 0-Framed*)
Timing[Z = DisjUnion@{Rp[5, 1], Rp[2, 6], Rp[7, 3], ur[4], Km[8], Km[9], Km[10]};
 For[i = 2, i ≤ 10, i++, Z = Z // m[1, i, 1]];
 Z // Expand // Together // Factor]
```

$\left\{5.29461,\ \left\{\frac{1 - t_1 + t_1^2}{t_1},\ 0,\ 0,\right.\right.$
$\left.\left. -\frac{1}{t_1^4}\left(1 - t_1 + t_1^2\right)^2 \left(-2 c_1 + t_1 + 2 c_1 t_1 - 2 t_1^2 + 3 t_1^3 - 2 c_1 t_1^3 - 2 t_1^4 + 2 c_1 t_1^4 + 2 a_1 w_1 + 2 a_1 t_1^3 w_1\right)\right\}\right\}$

$$-\frac{1}{t_1^4}\left(1 - t_1 + t_1^2\right)^2 \left(-2 c_1 + t_1 + 2 c_1 t_1 - 2 t_1^2 + 3 t_1^3 - 2 c_1 t_1^3 - 2 t_1^4 + 2 c_1 t_1^4 + 2 a_1 w_1 + 2 a_1 t_1^3 w_1\right) \;/.$$

```
   {c_ → 0, a_ → 0} // Expand // Exponent[#, t1] &
```

4

```
(*4_1 0-Framed*)
Timing[Z = DisjUnion@{Rm[8, 1], Rp[5, 9], Rm[2, 6], Rp[10, 3], nr[7], ur[4]};
 For[i = 2, i ≤ 10, i++, Z = Z // m[1, i, 1]];
 Z // Expand // Together // Factor]
```

$\left\{3.22325,\ \left\{-\frac{1 - 3 t_1 + t_1^2}{t_1},\ 0,\ 0,\ -\frac{1}{t_1^4}\left(1 + t_1\right)\left(1 - 3 t_1 + t_1^2\right)^3 \left(1 - 2 c_1 - t_1 + 2 c_1 t_1 + 2 a_1 w_1\right)\right\}\right\}$

$\frac{1}{t_1^4}(1+t_1)\left(1-3t_1+t_1^2\right)^3\left(1-2c_1-t_1+2c_1t_1+2a_1w_1\right)$ /. {c_ → 0, a_ → 0} // Expand

$\frac{1}{t_1^4}-\frac{9}{t_1^3}+\frac{29}{t_1^2}-\frac{36}{t_1}+36t_1-29t_1^2+9t_1^3-t_1^4$

```
(*5_1 0-framed*)
Timing[Z = DisjUnion@{Rp[7, 1], Rp[2, 8], Rp[9, 3],
    Rp[4, 10], Rp[11, 5], ur[6], Km[12], Km[13], Km[14], Km[15], Km[16]};
 For[i = 2, i ≤ 16, i++, Z = Z // m[1, i, 1]];
 Z // Expand // Together // Factor]
```

{92.7456,

$\left\{\frac{1-t_1+t_1^2-t_1^3+t_1^4}{t_1^2}, 0, 0, -\frac{1}{t_1^8}\left(1-t_1+t_1^2-t_1^3+t_1^4\right)^2\left(-4c_1+t_1+6c_1t_1-2t_1^2-6c_1t_1^2+4t_1^3+4c_1t_1^3-6t_1^4+8t_1^5-4c_1t_1^5-8t_1^6+6c_1t_1^6+7t_1^7-6c_1t_1^7-4t_1^8+4c_1t_1^8+4a_1w_1-2a_1t_1w_1+4a_1t_1^2w_1+4a_1t_1^5w_1-2a_1t_1^6w_1+4a_1t_1^7w_1\right)\right\}\}$

$-\frac{1}{t_1^8}\left(1-t_1+t_1^2-t_1^3+t_1^4\right)^2\left(-4c_1+t_1+6c_1t_1-2t_1^2-6c_1t_1^2+4t_1^3+4c_1t_1^3-6t_1^4+8t_1^5-4c_1t_1^5-8t_1^6+6c_1t_1^6+7t_1^7-6c_1t_1^7-4t_1^8+4c_1t_1^8+4a_1w_1-2a_1t_1w_1+4a_1t_1^2w_1+4a_1t_1^5w_1-2a_1t_1^6w_1+4a_1t_1^7w_1\right)$ /.
{c_ → 0, a_ → 0} // Expand // Exponent[#, t_1] &

8

Strand doubling

```
DoubleStrand[i_, J_, K_][E_] :=
 (*Just the q-coproduct. But we need to reorder the c-contributions to
   the quadratic part in the exponent giving rise to some perturbations*)
 Block[{C, P, Q, L, α, β, δ, j, k},

  C = E[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = E[[2]]; (*The linear term of the exponential*)
  Q = E[[3]]; (*The quadratic term, scaled by C.*)
  P = E[[4]]; (*Perturbation, scaled by C^4*)

  α  = C^-1 D[Q, w_i] /. {a_i → 0};
  β  = C^-1 D[Q, a_i] /. {w_i → 0};
  δ  = C^-1 D[Q, a_i, w_i];

  (*First add the necessary perturbations to P*)
  P = P +
    C^4 (α c_k w_j + α^2/2 w_j w_k - β a_k c_j - β^2/2 t_k a_k a_j - δ a_k c_j (w_j + w_k) - δ^2/2 t_k a_k a_j (w_j + w_k)^2 +
       δ (t_k a_j + a_k) c_k w_j + δ^2/2 (t_k a_j + a_k)^2 w_j w_k);
  (*Finally apply the obvious part of the coproduct, without ϵ.*)
  {C, L, Q, P} /. {a_i → t_K a_J + a_K, c_i → c_J + c_K, w_i → w_J + w_K, t_i → t_J t_K,
    a_j → a_J, a_k → a_K, c_j → c_J, c_k → c_K, w_j → w_J, w_k → w_K, t_j → t_J, t_k → t_K}
  (*Careful with the case i=J=K!*)
 ]
```

```
(*Test by doubling the overstrand of a positive crossing*)
Simplify[DoubleStrand[1, 4, 5]@Rp[1, 2] -
   (DisjUnion[{Rp[4, 2], Rp[5, x]}] // m[2, x, 2]) // Expand, Assumptions → {t_4 > 0}]
(*Test by doubling the understrand of a positive crossing*)
Simplify[DoubleStrand[2, 4, 5]@Rp[1, 2] -
   (DisjUnion[{Rp[1, 5], Rp[x, 4]}] // m[1, x, 1]) // Expand, Assumptions → {t_4 > 0}]
(*Test by doubling the overstrand of a negative crossing*)
Simplify[DoubleStrand[1, 4, 5]@Rm[1, 2] -
   (DisjUnion[{Rm[5, 2], Rm[4, x]}] // m[2, x, 2]) // Expand, Assumptions → {t_4 > 0}]
(*Test by doubling the understrand of a negative crossing*)
Simplify[DoubleStrand[2, 4, 5]@Rm[1, 2] -
   (DisjUnion[{Rm[1, 4], Rm[x, 5]}] // m[1, x, 1]) // Expand, Assumptions → {t_4 > 0}]
```

```
{0, 0, 0, 0}

{0, 0, 0, 0}

{0, 0, 0, 0}

{0, 0, 0, 0}
```

```
DoubleStrand[1, 4, 5]@Rm[1, 2]
```

$$\left\{1, -\text{Log}[t_4 t_5]\, c_2, -\frac{(a_5 + a_4 t_5)\, w_2}{t_4 t_5}, -c_2 (c_4 + c_5) + \frac{a_5 c_4 w_2}{t_4 t_5} + \frac{c_2 (a_5 + a_4 t_5)\, w_2}{t_4 t_5} - \frac{a_4 a_5 w_2^2}{2 t_4^2 t_5} - \frac{(a_5 + a_4 t_5)^2 w_2^2}{4 t_4^2 t_5^2}\right\}$$

```
(*Test by doubling a negative kink*)
(FullSimplify[
   (DisjUnion@{Rm[1, -3], Rm[2, 3], Rm[-1, -4], Rm[-2, 4], ur[-5], ur[5]} // m[1, 2, 1] //
           m[1, 5, 1] // m[1, 3, 1] // m[1, 4, 1] // m[-1, -2, -1] //
         m[-1, -5, -1] // m[-1, -3, -1] // m[-1, -4, -1] // Together) -
    DoubleStrand[1, -1, 1]@Km[1], Assumptions → {t₁ > 0, t₋₁ > 0}] // Together)
```

{0, 0, 0, 0}

```
(*Older Normal ordering operators*)
NO[w_i_, a_j_, k_][Z_] := Block[
  {α, β, δ, C, L, Q, P, QRest, μ, q},
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = μ^-1 ((1 - t_k) α β + α w_i + β a_j + δ a_j w_i);
  P = DP_{w_i→D_α, a_j→D_β}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α = C^-1 D[Q, w_i] /. a_j → 0;
  β = C^-1 D[Q, a_j] /. w_i → 0;
  δ = C^-1 D[Q, a_j, w_i];
  QRest = Q /. {a_j → 0, w_i → 0};
  μ = 1 - (1 - t_k) δ;

  Q = μ QRest + C * ((1 - t_k) α β + α w_i + β a_j + δ a_j w_i);
  P = μ^4 P - 1/4 (1 + t_k) C^4 Logos[α, β, δ, μ, a_j, c_k, w_i, t_k];
  C = μ C;
  {C, L, Q, P} // Together (*/.{a_{j|i}→a_k, c_{j|i}→c_k, w_{j|i}→w_k, t_{j|i}→t_k}*)
 ]

NO[w_k_, c_j_][Z_] := Block[
  {α, γ, C, L, Q, P, QRest, q},
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential*)
  Q = Z[[3]]; (*The quadratic term.*)
  P = Z[[4]];
  q = e^γ α w_k + γ c_j;
  P = DP_{w_k→D_α, c_j→D_γ}[P][Exp[q]] Exp[-q] // Expand; (*First deal with P formally*)
  Print["Formal P=", P];
  (*Then set the variables*)
  α = C^-1 D[Q, w_k];
```

```
  γ = D[L, c_j];
  Print[α, γ];

  QRest = Q /. {w_k → 0};

  Q = QRest + C e^γ α w_k;
  {C, L, Q, P} // Expand
 ]

NO[c_i_, a_k_][Z_] := Block[
  {β, γ, C, L, Q, P, QRest, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential*)
  Q = Z[[3]]; (*The quadratic term.*)
  P = Z[[4]];
  q = e^γ β a_k + γ c_i;
  P = DP_{a_k→D_β, c_i→D_γ}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  β = C^-1 D[Q, a_k];
  γ = D[L, c_i];

  QRest = Q /. {a_k → 0};

  Q = QRest + C e^γ β a_k;
  {C, L, Q, P} /. {b → Log[t]} // Expand
 ]

(*Needs debugging!*)
ReverseStrand[i_][E_] :=
 (*Just the q-coproduct. But we need to reorder the c-contributions to
   the quadratic part in the exponent giving rise to some perturbations*)
 Block[{C, P, Q, L, α, β, δ, x},
  C = E[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = E[[2]]; (*The linear term of the exponential*)
  Q = E[[3]]; (*The quadratic term, scaled by C.*)
  P = E[[4]]; (*Perturbation, scaled by C^4*)

  α = C^-1 D[Q, w_i] /. {a_i → 0};
  β = C^-1 D[Q, a_i] /. {w_i → 0};
  δ = C^-1 D[Q, a_i, w_i];

  (*Now apply the 0-co antipode*)
  C = C /. {a_i → -t_i^-1 a_i, c_i → -c_i, w_i → -w_i, t_i → t_i^-1};
  L = L /. {a_i → -t_i^-1 a_i, c_i → -c_i, w_i → -w_i, t_i → t_i^-1};
  Q = Q /. {a_i → -t_i^-1 a_i, c_i → -c_i, w_i → -w_i, t_i → t_i^-1};
  P = P /. {a_i → -t_i^-1 a_i, c_i → -c_i, w_i → -w_i, t_i → t_i^-1};
  Print["Q=", Q, " P=", P];

  (*Add the necessary perturbations to P coming from the 1-co S on Q.*)
  P = P +
```

```
    C⁴ (α wᵢ cᵢ + α²/2 wᵢ² - β tᵢ⁻¹ cᵢ aᵢ - β²/2 tᵢ⁻² aᵢ²);
  Print["α = " α, " P=", P];
  (*Do the δ terms really cancel? things are in anti-canonical order now!*)

  (*Finally reverse the order,
  using dummy variable k to avoid confusing the c's produced by NOwa.*)
  Print[{C, L, Q, P} // NO[wᵢ, cᵢ]];
  (({C, L, Q, P} // NO[wᵢ, cᵢ] // NO[wᵢ, aᵢ, x] // NO[cᵢ, aᵢ]) /.
     {aₓ → aᵢ, cₓ → cᵢ, wₓ → wᵢ, tₓ → tᵢ}) // Together
]
```

```
(*Testing Reversing overstrand of Rp gives negative crossing,
i.e. (S tensor id) (Rp) = Rm*)
Simplify[ReverseStrand[1]@Rp[1, 2] - Rm[1, 2], Assumptions → t₁ > 0]
```

$Q=-\frac{a_1 w_2}{t_1}$ $P=-c_1 c_2+\frac{a_1 c_1 w_2}{t_1}+\frac{a_1^2 w_2^2}{4 t_1^2}$

0 $P=-c_1 c_2-\frac{a_1^2 w_2^2}{4 t_1^2}$

00

$\left\{1, \text{Log}\left[\frac{1}{t_1}\right] c_2, -\frac{a_1 w_2}{t_1}, -c_1 c_2-\frac{a_1^2 w_2^2}{4 t_1^2}\right\}$

00

{0, 0, 0, 0}

```
Rp[1, 2]
```

$\left\{1, \text{Log}[t_1]\, c_2, a_1 w_2, c_1 c_2+a_1 c_1 w_2+\frac{1}{4} a_1^2 w_2^2\right\}$

```
(*Testing Reversing understrand of Rp gives negative crossing,
when ur and nr are added (id tensor S) (Rp) != Rm*)
DisjUnion@{ReverseStrand[2]@Rp[1, 2], ur[x], nr[y]} // m[y, 2, 2] // m[2, x, 2] // Together
Rm[1, 2] // Together
```

$Q = -a_1 w_2 \;\; P = -c_1 c_2 - a_1 c_1 w_2 + \frac{1}{4} a_1^2 w_2^2$

$\alpha \; = \; a_1 \;\; P = -c_1 c_2 - a_1 c_1 w_2 + a_1 c_2 w_2 + \frac{3}{4} a_1^2 w_2^2$

Formal $P = -c_1 c_2 + e^{\gamma} a_1 w_2 - e^{\gamma} \alpha c_1 w_2 - e^{\gamma} a_1 c_1 w_2 + e^{\gamma} a_1 c_2 w_2 + e^{2\gamma} \alpha a_1 w_2^2 + \frac{3}{4} e^{2\gamma} a_1^2 w_2^2$

$-a_1 - \text{Log}[t_1]$

$\left\{1, -\text{Log}[t_1]\, c_2, -\frac{a_1 w_2}{t_1}, -c_1 c_2 + \frac{a_1 w_2}{t_1} + \frac{a_1 c_2 w_2}{t_1} - \frac{a_1^2 w_2^2}{4 t_1^2}\right\}$

Formal $P = -c_1 c_2 + e^{\gamma} a_1 w_2 - e^{\gamma} \alpha c_1 w_2 - e^{\gamma} a_1 c_1 w_2 + e^{\gamma} a_1 c_2 w_2 + e^{2\gamma} \alpha a_1 w_2^2 + \frac{3}{4} e^{2\gamma} a_1^2 w_2^2$

$-a_1 - \text{Log}[t_1]$

$\left\{1, -\text{Log}[t_1]\, c_2, -\frac{a_1 w_2}{t_1}, \frac{1}{4 t_1^2}\left(-4 c_1 c_2 t_1^2 + 4 a_1 t_1 w_2 + 4 a_1 c_2 t_1 w_2 - a_1^2 w_2^2\right)\right\}$

$\left\{1, -\text{Log}[t_1]\, c_2, -\frac{a_1 w_2}{t_1}, \frac{1}{4 t_1^2}\left(-4 c_1 c_2 t_1^2 + 4 a_1 c_2 t_1 w_2 - a_1^2 w_2^2\right)\right\}$

Stability

Suppose we start with Z = {C,L,Q,P} such that deg(Q),deg(P)< deg(C) where deg refers to the degree in t or perhaps s=1-t and deg of a rational function f/g is defined as deg(f/g) = deg f - deg g. We call such Z stable (of degree deg C). It looks like stitching preserves this property: Given Z stable of degree d and Z' = $m^{ab}{}_c Z$ then Z' is again stable of degree d.

Below we test this idea on some generic examples. For this we only need the $t_i = t$, one variable version of the stitching formula. We call it m1

```
(*Stitching operation*)
NOwa1[i_, j_, k_][Z_] := Block[
  {α, β, δ, C, L, Q, P, QRest, μ, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = μ^-1 ((1 - t) α β + α w_i + β a_j + δ a_j w_i);
  P = DP_{w_i→D_α, a_j→D_β}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α  = C^-1 D[Q, w_i] /. a_j → 0;
  β  = C^-1 D[Q, a_j] /. w_i → 0;
  δ  = C^-1 D[Q, a_j, w_i];
  QRest = Q /. {a_j → 0, w_i → 0};
  μ  = 1 - (1 - t) δ;

  Q = μ QRest + C * ((1 - t) α β + α w_i + β a_j + δ a_j w_i);
  P = μ^4 P - 1/4 (1 + t) C^4 Logos[α, β, δ, μ , a_j, c_k, w_i, t];
  C = μ C;
  {C, L, Q, P} // Together
 ]

NOc1[i_, j_][Z_] := Block[
  {α, β, γ, Γ, δ, C, L, Q, P, QRest, q },
  C = Z[[1]]; (*The (inverse of the) constant in front, aka ω*)
  L = Z[[2]]; (*The linear term of the exponential, unscaled*)
  Q = Z[[3]]; (*The quadratic term, unscaled.*)
  P = Z[[4]];
  q = e^(Γ+γ) δ a_j w_i + e^Γ α w_i + e^γ β a_j + Γ c_j + γ c_i;
  P = DP_{w_i→D_α, c_j→D_Γ, a_j→D_β, c_i→D_γ}[P][Exp[q]] Exp[-q]; (*First deal with P formally*)
  (*Then set the variables*)
  α  = C^-1 D[Q, w_i] /. {a_j → 0};
  β  = C^-1 D[Q, a_j] /. {w_i → 0};
  δ  = C^-1 D[Q, a_j, w_i];
  γ  =  D[L, c_i];
  Γ =  D[L, c_j];
  QRest = Q /. {a_j → 0, w_i → 0};
  Q = QRest + C e^(Γ+γ) δ a_j w_i + C e^Γ α w_i + C e^γ β a_j;


  {C, L, Q, P} // Expand
 ]

m1[i_, j_, k_][Z_] := Block[{x},
  (Z // NOwa1[i, j, x] // NOc1[i, j]) /. {a_{i|j|x} → a_k, c_{i|j|x} → c_k, w_{i|j|x} → w_k} // Expand]
```

```
Z = {t^4 + t^3 + t^2 + 1, c3 Log[t] + c1 Log[t] + c2 Log[t],
   X11 a1 w1 + X12 a1 w2 + X21 a2 w1 + X22 a2 w2 + QOther , a1 t^2 + c2 w1};
Zp = Z // m1[1, 2, 1] // Together;
Zp[[3]]
Zp[[4]] /. {a_ → 0, c_ → 0} // (Exponent[Numerator[#], t] - Exponent[Denominator[#], t] &)
```

$$\frac{1}{1+t^2+t^3+t^4}\left(\text{QOther}+\text{QOther}\,t^2+\text{QOther}\,t^3+\text{QOther}\,t^4+t\,a_1 w_1 X_{11}+t^3 a_1 w_1 X_{11}+t^4 a_1 w_1 X_{11}+t^5 a_1 w_1 X_{11}+a_1 w_1 X_{12}+t^2 a_1 w_1 X_{12}+t^3 a_1 w_1 X_{12}+t^4 a_1 w_1 X_{12}-\text{QOther}\,X_{21}+\text{QOther}\,t\,X_{21}+t^2 a_1 w_1 X_{21}+t^4 a_1 w_1 X_{21}+t^5 a_1 w_1 X_{21}+t^6 a_1 w_1 X_{21}-a_1 w_1 X_{12} X_{21}+t\,a_1 w_1 X_{12} X_{21}+t\,a_1 w_1 X_{22}+t^3 a_1 w_1 X_{22}+t^4 a_1 w_1 X_{22}+t^5 a_1 w_1 X_{22}+a_1 w_1 X_{11} X_{22}-t\,a_1 w_1 X_{11} X_{22}\right)$$

10